



LibreOffice
The Document Foundation

Version 4.0

Guide Calc

Chapitre 7

Formules et fonctions



Copyright

Ce document est Copyright © 2010–2012 par ses contributeurs tels que listés ci-dessous. Vous pouvez le distribuer et/ou le modifier sous les termes des licences GNU General Public License (<http://www.gnu.org/licenses/gpl.html>), version 3 ou ultérieure ou Creative Commons Attribution License (<http://creativecommons.org/licenses/by/3.0/>), version 3.0 ou ultérieure.

Tous les noms de marque à l'intérieur de ce guide appartiennent à leur propriétaire légitime.

Contributeurs

Auteur : Christian Chenal

Relecteurs : François Bégasse, Philippe Clément, Pierre-Yves Samyn

Retours

Veuillez envoyer vos commentaires ou suggestions à propos de ce document à :
doc@fr.libreoffice.org

Remerciements

Ce chapitre est basé sur les Chapitres 7 de *OpenOffice.org 3.3 Calc Guide (anglais)*, *LibreOffice 3.4 Calc Guide (anglais)* et *LibreOffice 3.6 Guide Calc (français)*. Les contributeurs à ces chapitres sont :

Martin Fox	Kirk Abbott	Bruce Byfield	Stigant Fyrwitful
Barbara M. Tobias	John Viestenz	Claire Wood	Jean Hollis Weber
Barbara Duprey	Hal Parker	John A Smith	Christian Chenal
Philippe Clément	Pierre-Yves Samyn	Laurent Balland-Poirier	

Date de publication et version du logiciel

Publié le 13 juin 2013. Basé sur LibreOffice 4.0.0.

Apparence de LibreOffice

LibreOffice fonctionne sous les systèmes d'exploitation Windows, Linux et Mac OS X, chacun de ces systèmes pouvant exister sous plusieurs versions. Chacune de ces configurations peut donner lieu à des options d'affichage différentes (polices, couleurs, thèmes, gestionnaire de fenêtres, icônes). De plus, ces options d'affichage sont pour certaines personnalisables par les utilisateurs.

Les illustrations de ce chapitre ont été réalisées sous Windows avec le jeu d'icônes Galaxy. Certaines images ne ressembleront donc pas exactement à ce que vous voyez sur votre ordinateur. En cas de confusion sur une icône, dont l'apparence peut être très différente d'un jeu à l'autre, vous pouvez vérifier son nom, qui apparaît dans l'infobulle lorsque vous placez le pointeur de la souris au-dessus.

Note pour les utilisateurs Mac

Certaines combinaisons de touches et certains éléments de menus sont différents entre un Mac et Windows ou Linux. La table ci-dessous donne quelques équivalents pour les instructions de ce chapitre. Pour une liste plus détaillée, voyez dans l'Aide.

<i>Windows/Linux</i>	<i>Équivalent Mac</i>	<i>Effet</i>
Sélection de menu Outils > Options	LibreOffice > Préférences	Accès aux options de paramétrage
<i>Clic droit</i>	⌘+clic	Ouvre un menu contextuel
<i>Ctrl (Contrôle)</i>	⌘ (Commande)	Utilisé avec d'autres touches
<i>F5</i>	Maj+⌘+F5	Ouvre le Navigateur
<i>F11</i>	⌘+T	Ouvre la fenêtre Styles & Formatage

Table des matières

Copyright.....	2
Apparence de LibreOffice.....	2
Note pour les utilisateurs Mac.....	3
Introduction.....	6
Créer des formules.....	6
Opérateurs dans les formules.....	6
Types d'opérateurs.....	8
Opérateurs arithmétiques.....	8
Opérateurs de comparaison.....	8
Opérateurs de texte.....	9
Opérateurs de référence.....	11
Références relatives et absolues.....	13
Référence relative.....	13
Référence absolue.....	14
Ordre de calcul.....	16
Feuilles liées par des calculs.....	16
Calcul des formules.....	19
Comprendre les fonctions.....	20
Comprendre la structure des fonctions.....	20
Fonctions imbriquées.....	21
Liste de fonctions.....	21
Assistant Fonction.....	22
Ligne de saisie.....	24
Stratégies pour créer un classeur.....	25
Le piège des valeurs fixes.....	25
Manque de documentation.....	25
Formules de contrôle d'erreur.....	25
Stratégies pour créer des formules et des fonctions.....	26
Placer une formule unique dans chaque cellule.....	27
Séparer une formule en plusieurs parties, puis les combiner.....	27
Utiliser l'éditeur Basic pour créer des fonctions.....	27
Détecter et corriger les erreurs.....	27
Messages d'erreur.....	28
Exemples d'erreurs fréquentes.....	28
#DIV/0 ! (532) Division par zéro.....	28
#VALEUR ! Valeur incorrecte et #REF ! Référence incorrecte.....	30
Codage en couleur.....	30
L'Audit.....	31
Exemples de fonctions.....	33
Fonctions de base arithmétiques et statistiques.....	33
Arithmétique de base.....	33
Statistiques de base.....	34
Utiliser ces fonctions.....	35

<i>Arrondir les nombres.....</i>	<i>35</i>
<i>Méthodes d'arrondi.....</i>	<i>35</i>
<i>Utiliser les caractères génériques dans les fonctions.....</i>	<i>37</i>
<i>Fonctions avancées.....</i>	<i>38</i>

Introduction

Dans les chapitres précédents, nous avons saisi dans chaque cellule l'un des deux types de données de base : les nombres et les textes. Cependant, nous ne savons pas toujours ce que le contenu devrait être. Souvent, le contenu d'une cellule dépend du contenu d'autres cellules. Pour gérer cette situation, nous allons utiliser un troisième type de données : les *formules*. Les formules sont des équations qui utilisent des nombres, des textes et des variables pour obtenir un résultat. Dans un classeur, les variables sont des emplacements de cellules qui contiennent des données nécessaires à la résolution de l'équation.

Une *fonction* est un calcul prédéfini saisi dans une cellule pour vous aider à analyser ou à manipuler les données dans un classeur. Tout ce que vous avez à faire est d'ajouter les arguments, et le calcul sera automatiquement effectué pour vous. Les fonctions vous aident à créer les formules nécessaires pour obtenir les résultats que vous recherchez.

Créer des formules

Vous pouvez saisir des formules de deux manières, directement dans la cellule elle-même ou dans la ligne de saisie. Dans les deux cas, vous devez commencer une formule avec l'un des symboles suivants : =, + ou -. Si vous commencez par quelque chose d'autre, la formule sera considérée comme étant un texte simple.

Opérateurs dans les formules

Toutes les cellules d'une feuille peuvent être utilisées pour contenir des données ou le résultat de calculs. L'entrée de données s'effectue simplement en saisissant dans la cellule et en se déplaçant vers la cellule suivante, ou en appuyant sur *Entrée*. Pour les formules, le signe égal indique que la cellule sera utilisée pour un calcul. Un calcul mathématique comme $15 + 46$ peut être effectué comme indiqué Figure 1.

Si le calcul sur la gauche est effectué dans une seule cellule, la vraie puissance des formules est montrée sur la droite, où les données sont placées dans des cellules et le calcul est effectué en faisant référence à ces cellules. Dans ce cas, les cellules B3 et B4 contiennent les données, et B5 est la cellule où le calcul est effectué. Notez que la formule s'affiche sous la forme **=B3+B4**. Le signe plus indique que les contenus des cellules B3 et B4 sont additionnés et donc que le résultat est inséré dans la cellule contenant la formule. Toutes les formules sont construites suivant ce principe. D'autres façons de saisir des formules sont indiquées Tableau 1.

Ces références de cellules permettent aux formules d'utiliser les données où qu'elles soient dans la feuille en cours, ou dans d'autres feuilles du classeur qui est ouvert. Si les données nécessaires sont dans des feuilles différentes, elles seront référencées en utilisant le nom de la feuille, par exemple **=SOMME(Feuille2.B12+Feuille3.A11)**.

Note

Pour saisir le symbole = pour un autre propos que celui de créer une formule comme décrit dans ce chapitre, tapez une apostrophe avant ce symbole. Par exemple, dans la saisie '= peut signifier plusieurs choses pour des personnes différentes, Calc traite tout ce qui se trouve après l'apostrophe, et donc le signe =, comme du texte.

Calcul simple dans une cellule				Calcul par référence			
	A	B	C		A	B	C
1				1			
2				2			
3		=15+46		3		15	
4				4		46	
5				5		=B3+B4	
6				6			

	A	B	C		A	B	C
1				1			
2				2			
3			61	3		15	
4				4		46	
5				5		61	
6				6			

Figure 1 : Un calcul simple

Tableau 1. Façons ordinaires de saisir des formules

Formule	Description
=A1+10	Affiche le contenu de la cellule A1 plus 10.
=A1*16%	Affiche 16 % du contenu de la cellule A1.
=A1*A2	Affiche le résultat de la multiplication des cellules A1 et A2.
=ARRONDI (A1;1)	Affiche le contenu de la cellule A1 arrondi à une décimale.
=TAUX.EFFECTIF (5%;12)	Calcule le taux d'intérêt effectif pour un taux d'intérêt nominal de 5 % et 12 paiements par an.
=B8 - SOMME (B10 : B14)	Calcule B8 moins la somme des cellules de B10 à B14.
=SOMME (B8 ; SOMME (B10 : B14))	Calcule la somme des cellules B10 à B14 et ajoute la valeur de B8.
=SOMME (B1 : B1048576)	Fait la somme de tous les nombres de la colonne B.
=MOYENNE (Glucide)	Affiche la moyenne de la plage nommée Glucide.
=SI (C31>140; "HAUT"; "OK")	Affiche le résultat de l'analyse conditionnelle de deux données. Si le contenu de C31 est supérieur à 140, alors HAUT est affiché, et sinon OK est affiché.

Note

Les utilisateurs de Lotus 1-2-3®, de Quattro Pro® ou d'autres tableurs sont peut-être habitués à des formules qui commencent par +, -, =, (, @, ., \$ ou #. Les formules mathématiques peuvent ressembler à **+D2+C2** ou à **+2*3**. Les fonctions commencent par le symbole @ comme **@SUM (D2 . . D7)**, **@COS (@DEGTORAD (30))** et **@IRR (GUESS ; CASHFLOWS)**. Les plages sont représentées comme A1..D3.

Les fonctions peuvent être identifiées dans le Tableau 1 par un mot, par exemple ARRONDI, suivi de parenthèses renfermant des références ou des nombres.

Il est également possible de définir des plages en les nommant, par le choix de **Insertion > Noms**. Par exemple, Glucide représente une plage comme B3:B10. Des fonctions de logique peuvent aussi être mises en œuvre, comme l'expression SI qui aura comme résultat une réponse conditionnelle selon la donnée dans la cellule identifiée, par exemple :

=SI(A2>0;"Positif";"Négatif")

Une valeur de 3 dans la cellule A2 retournera le résultat *Positif*, -9 le résultat *Négatif*.

Types d'opérateurs

Vous pouvez utiliser les types d'opérateurs suivants dans LibreOffice Calc : arithmétiques, de comparaison, de texte et de référence.

Opérateurs arithmétiques

Les opérateurs d'addition, de soustraction, de multiplication et de division retournent des résultats numériques. Les opérateurs de négation et de pourcentage identifient une caractéristique du nombre qui se trouve dans la cellule, par exemple -37. L'exemple pour l'exponentiation illustre comment saisir un nombre qui est multiplié par lui-même un certain nombre de fois, par exemple $2^3 = 2 \times 2 \times 2$.

Tableau 2. Opérateurs arithmétiques

Opérateur	Nom	Exemple
+ (Plus)	Addition	=1+1
– (Moins)	Soustraction	=2–1
– (Moins)	Négation	–5
* (astérisque)	Multiplication	=2*2
/ (barre oblique)	Division	=10/5
% (Pour cent)	Pourcentage	=15%
^ (accent circonflexe)	Exponentiation	=2^3

Opérateurs de comparaison

Les opérateurs de comparaison se rencontrent dans des formules qui utilisent la fonction SI et retournent une réponse soit vraie soit fausse ; par exemple, **=SI(B6>G12; 127; 0)** signifie que si le contenu de la cellule B6 est supérieur au contenu de la cellule G12, Calc va retourner le nombre 127, ou sinon Calc va retourner le nombre 0.

Une réponse directe VRAI ou FAUX peut être obtenue en saisissant une formule comme **=B6>G12**. Si les nombres trouvés dans les cellules référencées sont corrects et que celui de la cellule B6 est supérieur à celui de la cellule G12, la réponse VRAI est retournée, et dans le cas contraire la réponse FAUX est retournée.

Tableau 3. Opérateurs de comparaison

Opérateur	Nom	Exemple
= (signe égal)	Égal	A1=B1
> (Supérieur à)	Supérieur à	A1>B1
< (Inférieur à)	Inférieur à	A1<B1
>= (Supérieur ou égal à)	Supérieur ou égal à	A1>=B1
<= (Inférieur ou égal à)	Inférieur ou égal à	A1<=B1
<> (Différence)	Différence	A1<>B1

Si la cellule A1 contient la valeur numérique 4 et la cellule B1 la valeur numérique 5, les exemples ci-dessus retourneront respectivement les résultats FAUX, FAUX, VRAI, FAUX, VRAI et VRAI.

Opérateurs de texte

Les utilisateurs placent souvent des textes dans les classeurs. Pour amener de la flexibilité sur la manière dont ce type de données sera affiché, les textes peuvent être assemblés à partir de fragments provenant de différents endroits du classeur, comme montré Figure 2.

The figure consists of two screenshots of an Excel spreadsheet. The top screenshot shows the formula bar with the formula `=B2 & ", "& B3 & " " & B4` and the spreadsheet data. The bottom screenshot shows the same spreadsheet with the result of the formula displayed in cell B5.

	A	B	C
1			
2	Numéro	13	
3	Rue	avenue du Pdt Wilson	
4	Commune	Paris 16	
5	Adresse	=B2 & ", "& B3 & " " & B4	
6			

	A	B	C
1			
2	Numéro	13	
3	Rue	avenue du Pdt Wilson	
4	Commune	Paris 16	
5	Adresse	13, avenue du Pdt Wilson Paris 16	
6			

Figure 2 : Concaténation de texte

Dans cet exemple, des morceaux de texte particuliers se trouvent dans différentes cellules. Pour assembler ces segments, vous devez ajouter à la formule les espaces et ponctuations nécessaires entre guillemets, d'où la formule résultante `=B2 & ", "& B3 & " " & B4`. Le résultat est la concaténation en un texte représentant une adresse complète.

Calc dispose d'une fonction CONCATENER qui fait la même opération.

Pour prolonger cet exemple, si la cellule de résultat est définie en tant que nom, alors une concaténation du texte en utilisant ce nom défini peut être effectuée. Cette méthode est illustrée, Figure 3, Figure 4 et Figure 5, où la cellule contenant l'adresse complète est nommée "AdresseC" et est ensuite utilisée dans la formule d'une autre cellule.




Figure 3 : Définir un nom pour une plage




Figure 4 : Nommer une cellule ou une plage pour utilisation dans une formule

SOMME					=B8&AdresseC				
	A	B				C			
1									
2	Numéro	13							
3	Rue	avenue du Pdt Wilson							
4	Commune	Paris 16							
5	Adresse	13, avenue du Pdt Wilson Paris 16							
6									
7									
8		L'exposition aura lieu							
9									
10		=B8&AdresseC							
11									

	A	B				C		D
1								
2	Numéro	13						
3	Rue	avenue du Pdt Wilson						
4	Commune	Paris 16						
5	Adresse	13, avenue du Pdt Wilson Paris 16						
6								
7								
8		L'exposition aura lieu						
9								
10		L'exposition aura lieu 13, avenue du Pdt Wilson Paris 16						
11								

Figure 5 : Utilisation du nom défini dans une formule

Opérateurs de référence

Dans sa forme la plus simple, une référence se réfère à une seule cellule, mais elle peut aussi se référer à une plage rectangulaire ou à trois dimensions, ou encore à une autre référence dans une liste. Pour construire de telles références, vous avez besoin des opérateurs de référence.

Une cellule individuelle est identifiée par l'identifiant de sa colonne (lettre), situé en haut des colonnes, et par l'identifiant de sa ligne (nombre), situé du côté gauche du classeur. Ainsi, dans les classeurs, la cellule supérieure gauche est A1.

Opérateur de plage

L'opérateur de plage est représenté par le signe deux-points (:). Une expression utilisant l'opérateur de plage a la syntaxe suivante :

référence gauche : référence droite

L'opérateur de plage construit une référence à la plage comprise entre la référence gauche et la référence droite.

Dans le coin supérieur gauche de la Figure 6, la référence **A1:D12** est affichée, ce qui correspond aux cellules incluses dans l'opération. Il est nécessaire de faire glisser la souris de A1 en D12 pour mettre cette plage en surbrillance.




Figure 6 : Opérateur de référence pour une plage

Exemples :

A2:B4

Référence à une plage rectangulaire de 6 cellules, de 2 colonnes de large x 3 lignes de haut. Quand vous cliquez sur cette référence dans une formule dans la ligne de saisie, une bordure matérialise le rectangle.

Feuille1.A3:Feuille3.D4

Référence à une plage à trois dimensions de 24 cellules, de 4 colonnes de large x 2 lignes de haut x 3 feuilles de profondeur.

Si vous saisissez directement **B4:A2** ou **A4:B2**, Calc le transformera alors en **A2:B4**. Ainsi, la cellule en haut à gauche sera à gauche des deux-points, et la cellule en bas à droite sera à droite des deux-points. Mais si vous nommez la cellule B4 **_debut** et la cellule A2 **_fin**, vous pouvez utiliser **_debut:_fin** sans provoquer d'erreur.

Calc ne peut pas référencer une colonne entière de longueur non spécifiée en utilisant **A:A** ou une ligne entière en utilisant **1:1**, comme vous avez peut-être l'habitude de le faire avec d'autres tableurs. Il faut utiliser **A1:A1048576** pour référencer toute la colonne A et **A1:AMJ1** pour référencer toute la ligne 1.

Note

Outils > Options > LibreOffice Calc > Formule > Syntaxe de la formule permet d'utiliser les notations *Excel* A1 ou *Excel* L1C1. Il est toutefois recommandé de garder la syntaxe par défaut *Calc* A1.

Opérateur de concaténation de références

L'opérateur de concaténation de références est représenté par le signe tilde (~). Une expression utilisant l'opérateur de concaténation de références a la syntaxe suivante :

référence gauche ~ référence droite

Le résultat d'une telle expression est une référence de liste, qui est une liste ordonnée de références. Quelques fonctions acceptent une référence de liste en argument, comme SOMME, MAX ou INDEX par exemple.

La concaténation de références est parfois appelée "union". Mais ce n'est pas l'union de deux ensembles "référence gauche" et "référence droite" telle qu'on l'entend dans la théorie des ensembles. **NB(A1:C3~B2:D2)** retourne la valeur 12 (9+3) si toutes les cellules contiennent un nombre, mais l'union des deux références contient uniquement 10 cellules.

Notez que **SOMME(A1:C3;B2:D2)** est différent de **SOMME(A1:C3~B2:D2)**, bien que les résultats soient les mêmes. La première expression est l'appel d'une fonction avec deux paramètres, chacun étant une référence à une plage. La seconde expression est l'appel d'une fonction avec un paramètre, qui est une référence de liste.

Opérateur d'intersection

L'opérateur d'intersection est représenté par le signe point d'exclamation (!). Une expression utilisant l'opérateur d'intersection a la syntaxe suivante :

référence gauche ! référence droite

Si les références sont des plages simples, le résultat est une référence à une plage simple, qui contient toutes les cellules se trouvant à la fois dans la référence gauche et dans la référence droite.

Si les références sont des références de liste, alors chaque élément de la liste de gauche est croisé avec chaque élément de la liste de droite, et ces résultats sont concaténés dans une référence de liste. L'ordre est tout d'abord de croiser le premier élément de la liste de gauche avec tous les éléments de la liste de droite, puis de croiser le deuxième élément de la liste de gauche avec tous les éléments de la liste de droite, et ainsi de suite.

Exemples :

A2:B4 ! B3:D6

Le résultat est une référence à la plage B3:B4, parce que ces cellules sont dans A2:B4 et dans B3:D6.

(A2:B4~B1:C2) ! (B2:C6~C1:D3)

Les intersections A2:B4!B2:C6, A2:B4!C1:D3, B1:C2!B2:C6 et B1:C2!C1:D3 sont tout d'abord calculées. Le résultat est B2:B4, vide, B2:C2 et C1:C2. Puis ces résultats sont concaténés, en supprimant les parties vides. Le résultat final est donc la référence de liste B2:B4 ~ B2:C2 ~ C1:C2.

Vous pouvez utiliser l'opérateur d'intersection pour faire référence à une cellule d'un tableau croisé d'une façon explicite. Si vous avez des colonnes intitulées Températures et Précipitations, et des lignes intitulées Janvier, Février, Mars et ainsi de suite, alors l'expression suivante

'Février' ! 'Température'

fera référence à la cellule contenant la température du mois de février.

L'opérateur d'intersection (!) devrait avoir une priorité plus haute que l'opérateur de concaténation (~), mais ne vous fiez pas aux priorités.

Références relatives et absolues

Une référence, c'est-à-dire la façon dont on se réfère à l'emplacement d'une cellule particulière dans Calc, peut être soit relative (à la cellule en cours), soit absolue (fixe).

Référence relative

La Figure 7 est un exemple de référence relative qui illustre la différence entre les références relatives et absolues dans les classeurs.

- 1) À partir de l'exemple de la Figure 1, tapez les nombres 4 et 11 respectivement dans les cellules C3 et C4.
- 2) Copiez la formule de la cellule B5 dans la cellule C5. Vous pouvez le faire en utilisant un simple copier-coller ou en cliquant sur la poignée de B5 et faisant glisser vers C5, comme ci-dessous. La formule de B5 calcule la somme des valeurs des cellules B3 et B4.
- 3) Cliquez dans la cellule C5. La barre de formule affiche **=C3+C4** au lieu de **=B3+B4**, et la valeur de C5 est 15, la somme de 4 et de 11 qui sont les valeurs dans C3 et C4.

Dans la cellule B5, les références aux cellules B3 et B4 sont des références relatives. Cela signifie que Calc interprète la formule de B5, l'applique aux cellules de la colonne B et dépose le résultat dans la cellule qui contient cette formule. Quand vous copiez la formule dans une autre cellule, la même procédure est appliquée pour calculer la valeur de cette nouvelle cellule. Cette fois, la formule dans la cellule C5 se réfère aux cellules C3 et C4.

	A	B	C	D
1				
2				
3		15	4	
4		46	11	
5		61		
6				

	A	B	C	D
1				
2				
3		15	4	
4		46	11	
5		61	15	
6				

Figure 7 : Références relatives

Vous pouvez comprendre une adresse relative comme étant une paire de déplacements depuis la cellule en cours. La cellule B1 est une colonne à gauche et quatre lignes au-dessus de la cellule C5. L'adresse pourrait être écrite L[-1]C[-4]. En fait, des tableurs plus anciens permettaient ce type de notation dans les formules. Vous pouvez utiliser ce type de syntaxe en allant dans **Outils > Options > LibreOffice Calc > Formule** et en choisissant comme syntaxe de la formule **Excel L1C1**. Dans ce cas, notez que les lignes sont représentées par une syntaxe de type R1 (héritage de la langue anglaise), et non L1 comme son intitulé le laisse supposer.

À chaque fois que vous copiez cette formule de B5 dans une autre cellule, le résultat sera toujours la somme des deux nombres pris dans les deux cellules une ligne et deux lignes au-dessus de la cellule qui contiendra la formule.

L'adressage relatif est la méthode par défaut pour se référer à des adresses dans Calc.

Référence absolue

Vous pouvez vouloir multiplier une colonne de nombres par une valeur fixe, une colonne de montants en Dollars US par exemple. Pour convertir ces montants en Euros, il est nécessaire de multiplier chaque montant en dollars par le taux de change. 10,00 USD sera multiplié par 0,75 pour être converti en euros, ce qui donne ici 7,50 EUR. L'exemple suivant montre comment entrer un taux de change et l'utiliser pour convertir une colonne de montants USD en EUR.

- 1) Entrez le taux de change EUR-USD (0,75) dans la cellule D1. Saisissez des montants (en USD) dans les cellules D2, D3 et D4, par exemple 10, 20 et 30.
- 2) Dans la cellule E2, saisissez la formule **=D2*D1**. Le résultat est 7,5, ce qui est correct.
- 3) Copiez la formule de la cellule E2 dans la cellule E3. Le résultat est 200, ce qui est faux évidemment ! Calc a copié la formule en utilisant l'adressage relatif. La formule dans E3 est **=D3*D2** et non pas ce qui est voulu, soit **=D3*D1**.
- 4) Dans la cellule E2, modifiez la formule en **=D2*\$D\$1**. Copiez la dans les cellules E3 et E4. Les résultats sont alors 15 et 22,5, ce qui est correct.

=	=D2*D1		
D	E	F	
0,75 €			
\$10,00	7,50		
\$20,00			
\$30,00			

Fixer le taux de change à 0,75, puis copier E2 en E3

=	=D3*D2		
D	E	F	
0,75 €			
\$10,00	7,50		
\$20,00	200,00		
\$30,00			

✓	=	=D2*\$D\$1	
D	E	F	
0,75 €			
\$10,00	=D2*\$D\$1		
\$20,00	200,00		
\$30,00			

Changer la formule pour passer en référence absolue

=	=D2*\$D\$1		
D	E	F	
0,75 €			
\$10,00	7,50		
\$20,00	200,00		
\$30,00			

=	=D3*\$D\$1		
D	E	F	
0,75 €			
\$10,00	7,50		
\$20,00	15,00		
\$30,00			

Appliquer la formule correcte de E2 à E3

Figure 8 : Références absolues

Le signe \$ avant le D et avant le 1 convertit la référence à la cellule D1 du type relatif au type absolu ou fixe. Si la formule est copiée dans une autre cellule, la seconde partie sera toujours \$D\$1. L'interprétation de cette formule est "prendre la valeur de la cellule une colonne à gauche dans la même ligne et la multiplier par la valeur de la cellule D1".

Les références de cellules peuvent être effectuées de quatre façons.

Référence	Explication
D1	Relative, depuis la cellule E3, c'est la cellule une colonne à gauche et deux lignes au-dessus
\$D\$1	Absolue, depuis la cellule E3, c'est la cellule D1
\$D1	Partiellement absolue, depuis la cellule E3, c'est la cellule dans la colonne D et deux lignes au-dessus
D\$1	Partiellement absolue, depuis la cellule E3, c'est la cellule une colonne à gauche et dans la ligne 1

Astuce

Pour changer de type de référence dans les formules, cliquez sur la référence de cellule et appuyez sur *Maj+F4* pour basculer entre les quatre différents types de référence. Le plus efficace est d'utiliser ce raccourci clavier au fur et à mesure, dès que vous avez entré une référence, sans re cliquer dans la formule.

La connaissance de l'usage des références relatives et absolues est essentielle si vous voulez copier et coller des formules et lier des feuilles de calcul.

Note

Les références aux feuilles d'un classeur sont également relatives ou absolues. Par défaut, la référence est relative. Pour utiliser une référence absolue, faites précéder le nom de la feuille par le signe \$. Voir le Chapitre 10 (Lier les données de Calc) pour plus d'informations sur la façon de référencer des feuilles.

Ordre de calcul

L'*ordre de calcul* se réfère à la séquence dans laquelle les opérations numériques sont effectuées. Les divisions et les multiplications sont effectuées avant les additions et les soustractions. Il existe une tendance naturelle à considérer que les calculs s'effectuent de gauche à droite, à l'instar d'une équation qui serait lue comme un texte en français. En réalité, Calc évalue la formule en entier, puis, conformément aux règles de priorité, va morceler les formules pour exécuter les multiplications et les divisions avant les autres opérations. Par conséquent, vous devriez tester votre formule pour vous assurer que le résultat correct est bien obtenu. Voici un exemple de l'ordre de calcul des opérations :

Table 4. Ordre de calcul

Calcul de gauche à droite	Calcul ordonné dans LibreOffice
1+3*2+3 = 11 1 + 3 = 4, puis 4 x 2 = 8, puis 8 + 3 = 11	=1+3*2+3 donne 10 3 x 2 = 6, puis 1 + 6 + 3 = 10
Une autre intention pourrait être : 1+3*2+3 = 20	Le programme résout la multiplication 3 x 2 avant de s'occuper des nombres à ajouter.
6/2*3 = 1 2 x 3 = 6, puis 6 / 6 = 1	6/2*3 donne 9 À niveau de priorité égal, le programme effectue les opérations de gauche à droite.

Si vous voulez l'un des résultats sur la gauche, la manière de les obtenir est d'ordonner la formule :

$$((1+3) * 2)+3 = 11 \quad (1+3) * (2+3) = 20 \quad 6/(2*3) = 1$$

Note

Utilisez des parenthèses pour regrouper les opérations dans l'ordre que vous souhaitez ; par exemple, **=B4+G12*C4/M12** peut devenir **=((B4+G12) *C4) /M12**. L'utilisation des parenthèses permet aussi de rendre plus facilement compréhensibles les formules les plus complexes.

Feuilles liées par des calculs

Une autre puissante fonctionnalité de Calc est la possibilité de lier des données de plusieurs feuilles. Nommer les feuilles avec pertinence peut être utile pour identifier quelles sont les données employées. Un nom comme Salaires ou Ventes Sud, par exemple, est beaucoup plus significatif que Feuille1. La fonction FEUILLE() retourne le numéro d'ordre de la feuille dans une collection de feuilles de classeur. Ce numéro d'ordre peut ne pas correspondre aux noms par défaut Feuille1, Feuille2, et ainsi de suite, si les feuilles sont ensuite déplacées à des endroits différents.

Pour illustrer comment obtenir des données d'autres feuilles, voici un exemple de tableau professionnel où une entreprise cumule les chiffres d'affaires et les coûts de ses magasins opérationnels sur une seule feuille de synthèse.

3	Ventes cumulées annuelles				
4					
5					
6		Oct	Nov	Dec	Année
7	Ventes :				
8	Plantes	36 288	52 874	81 335	1 283 107
9	Engrais	16 822	3 825	3 600	697 634
10	Terreux	2 019	459	432	84 479
11	Sous-total	55 129	57 158	85 367	2 065 220
12					
13	Coût des ventes :				
14	Fournitures	18 744	19 434	29 025	702 175
15	Taxes sur ventes	6 064	6 287	9 390	227 174
16	Sous-total	24 808	25 721	38 415	929 349
17					
18	CA total	30 321	31 437	46 952	1 135 871
19					
20	Erre				

3	Ventes cumulées annuelles				
4					
5					
6		Oct	Nov	Dec	Année
7	Ventes :				
8	Plantes	38 251	14 899	49 588	1 027 538
9	Engrais	6 120	2 384	7 934	164 406
10	Terreux	734	286	952	19 729
11	Sous-total	45 105	17 569	58 474	1 211 673
12					
13	Coût des ventes :				
14	Fournitures	15 336	5 973	19 881	411 969
15	Taxes sur ventes	4 962	1 933	6 432	133 284
16	Sous-total	20 298	7 906	26 313	545 253
17					
18	CA total	24 807	9 663	32 161	666 420
19					
20	Erie				

Magasin1
Magasin2
Magasin3
Cumul
+

3	Ventes cumulées annuelles				
4					
5					
6		Oct	Nov	Dec	Année
7	Ventes :				
8	Plantes	65 801	58 257	102 179	1 498 444
9	Engrais	54 833	17 620	8 782	843 175
10	Terreaux	59 025	16 824	7 622	397 342
11	Sous-total	179 659	92 701	118 583	2 738 961
12					
13	Coût des ventes :				
14	Fournitures	61 084	31 518	40 318	931 247
15	Taxes sur ventes	19 762	10 197	13 044	301 286
16	Sous-total	80 846	41 715	53 362	1 232 533
17					
18	CA total	98 813	50 986	65 221	1 506 428
19					
20	Frais				

Feuille
contenant
les
données
pour le
magasin 3

K8					=	=Magasin1.K8+Magasin2.K8+Magasin3.K8			
	A	K	L	M	N				
1									
2									
3	Ventes cumulées annuelles								
4									
5									
6		Oct	Nov	Dec	Année				
7	Ventes :								
8	Plantes	140 340	126 030	233 102	3 809 089				
9	Engrais	77 775	23 829	20 316	1 705 215				
10	Terreaux	61 778	17 569	9 006	501 550				
11	Sous-total	279 893	167 428	262 424	6 015 854				
12									
13	Coût des ventes :								
14	Fournitures	95 164	56 925	89 224	2 045 391				
15	Taxes sur ventes	30 788	18 417	28 866	661 744				
16	Sous-total	125 952	75 342	118 090	2 707 135				
17									
18	CA total	153 941	92 086	144 334	3 308 719				
19									
20	Frais								



Magasin1 / Magasin2 / Magasin3

Cumul 



Feuille
contenant
les
données
cumulées
de tous les
magasins

Figure 9 : Combiner les données de plusieurs feuilles dans une seule feuille

Les feuilles ont été construites avec des structures identiques. La manière la plus facile de le faire est la suivante ;

- 1) Élaborez la feuille du premier magasin, saisissez les données, formatez les cellules et préparez les formules pour les différentes sommes des lignes et des colonnes.
- 2) Faites un clic droit sur l'onglet de la feuille et choisissez **Renommer la feuille**. Saisissez **Magasin1**.
- 3) Faites à nouveau un clic droit sur l'onglet de la feuille et choisissez **Déplacer/Copier la feuille**. Dans la boîte de dialogue, cochez l'option *Copier* et sélectionnez - *placer en dernière position* - dans le champ *Insérer avant*. Dans la zone *Nouveau nom*, remplacez Magasin1_2 par **Magasin2**. Répétez l'opération pour produire les feuilles Magasin3 et Cumul.
- 4) Saisissez les données des magasins 2 et 3 dans leurs feuilles respectives. Chaque feuille est indépendante et rapporte les résultats des magasins individuels.
- 5) Dans la feuille Cumul, cliquez dans la cellule K8. Tapez =, sélectionnez l'onglet Magasin1, cliquez dans la cellule K8, appuyez sur +, faites de même pour les feuilles Magasin2 et Magasin3 et appuyez sur *Entrée*. Vous obtenez alors dans la cellule K8 la formule **=Magasin1.K8+Magasin2.K8+Magasin3.K8**, qui additionne les ventes de plantes pour les trois magasins.
- 6) Copiez la formule, sélectionnez la plage K8:N18, choisissez **Édition > Collage spécial**, décochez les cases *Tout insérer* et *Formats* dans la section *Sélection* de la boîte de dialogue et cliquez sur **OK**. Si un message d'avertissement apparaît, vous demandant si vous voulez écraser les données existantes, cliquez sur **Oui**.
- 7) Vous avez alors copié les formules dans chaque cellule tout en conservant le format que vous aviez paramétré dans la feuille de départ. Bien sûr, vous devrez remettre de l'ordre dans la feuille en supprimant les zéros des lignes non formatées.

Note

Le comportement par défaut de LibreOffice est de coller tous les attributs de la cellule d'origine (formats, notes, objets, chaînes de caractères et nombres).

L'Assistant fonction peut être également utilisé pour effectuer le lien. L'utilisation de cet Assistant est décrite plus en détail dans la section sur les Fonctions.

Calcul des formules

Par défaut, les formules se recalculent automatiquement. Vous pouvez changer ce comportement en allant dans **Outils > Contenu des cellules > Calculer automatiquement**. Dans ce cas, vous pouvez déclencher manuellement le recalcul des formules en choisissant **Outils > Contenu des cellules > Recalculer** ou en appuyant sur la touche *F9*. Ceci peut être utile si votre classeur comprend beaucoup de formules.

Vous pouvez aussi empêcher le recalcul automatique des formules à l'ouverture du classeur en allant dans la section *Recalcul au chargement du fichier* de **Outils > Options > LibreOffice Calc > Formule** et en choisissant le comportement souhaité pour les fichiers au format Excel 2007 et ses versions ultérieures d'une part, et pour les classeurs au format ODF non enregistrés avec LibreOffice d'autre part.

Comprendre les fonctions

Calc comprend plus de 350 fonctions pour vous aider à analyser et référencer les données. La plupart de ces fonctions sont destinées à être utilisées avec des nombres, mais un certain nombre d'autres sont utilisées avec des dates et heures, ou même des textes. Une fonction peut être aussi simple que l'addition de deux nombres, ou de calculer la moyenne d'une série de nombres. D'un autre côté, elle peut être aussi complexe que de calculer la variance d'un échantillon, ou la tangente hyperbolique d'un nombre.

Le nom d'une fonction est ordinairement une description abrégée de ce que fait cette fonction. Par exemple, la fonction VA calcule la valeur actuelle d'un investissement, et BINHEX convertit un nombre binaire en un nombre hexadécimal. Les fonctions sont affichées entièrement en lettres majuscules, même si elles ont été saisies en minuscules. Calc les lit correctement si elles sont saisies en minuscules ou avec une casse mélangée.

Attention



Par défaut, LibreOffice utilise les noms de fonction en français (SOMME par exemple). Vous pouvez travailler avec les noms en anglais (SUM par exemple) en sélectionnant **Outils > Options > LibreOffice Calc > Formule** et en cochant *Utiliser les noms de fonction en anglais*. Le fait de cocher ou décocher cette option va modifier le nom des fonctions existantes dans votre classeur.

Le *Guide Calc* en français utilise les noms de fonctions en français.

Quelques fonctions de base sont à peu près similaires aux opérateurs. Exemples :

- + Cet opérateur additionne deux nombres pour un résultat. SOMME() a également la capacité d'additionner des cellules ou des groupes de plages contenant des nombres.
- * Cet opérateur multiplie deux nombres pour un résultat. PRODUIT() est à la multiplication ce que SOMME() est à l'addition.

Chaque fonction comporte un nombre d'*arguments* utilisés pour les calculs. Ces arguments peuvent avoir leur propre nom, ou pas. Votre rôle est d'entrer les arguments nécessaires pour exécuter la fonction. Dans certains cas, les arguments ont des valeurs prédéfinies, et vous pouvez avoir besoin de vous référer à l'Aide en ligne ou à l'Annexe B (Description des fonctions) de ce livre pour les comprendre. Le plus souvent cependant, un argument est une valeur que vous saisissez manuellement, ou une valeur déjà saisie dans une cellule ou une plage de cellules dans la feuille. Dans Calc, vous pouvez référencer les valeurs d'autres cellules en entrant leur référence ou leur nom, ou en les sélectionnant avec la souris, sauf dans certains cas. Si les valeurs dans ces cellules changent, le résultat de cette fonction est automatiquement mis à jour.

Pour des raisons de compatibilité, dans Calc, les fonctions et leurs arguments ont pour la plupart des noms identiques à leur équivalent dans Microsoft Excel. Cependant, Excel et Calc peuvent avoir des fonctions qui ne sont pas forcément semblables. Parfois, les fonctions au nom identique entre Calc et Excel peuvent avoir des arguments différents, ou un même argument avec un nom légèrement différent. Dans ce cas, ces fonctions ne peuvent être importées d'un logiciel à l'autre. Cependant, la majorité des fonctions peuvent être utilisées à la fois dans Calc et dans Excel sans aucune modification.

Comprendre la structure des fonctions

Toutes les fonctions ont une structure similaire. Si vous utilisez le bon outil pour saisir une fonction, vous pouvez éviter d'apprendre sa structure, quoique ce soit toujours utile en cas de problème.

Pour donner un exemple type, la structure de la fonction pour compter toutes les cellules dont le contenu correspond aux critères de recherche est la suivante :

=BDNB(Base de données;Champ de base de données;Critères de recherche)

Parce qu'une fonction ne peut exister par elle-même, elle doit toujours faire partie d'une formule. Par conséquent, même si la formule n'utilise qu'une seule fonction, il doit toujours y avoir un signe = au début de la formule. Quel que soit l'endroit dans la formule où la fonction se trouve, elle doit commencer par son nom, comme BDNB dans l'exemple ci-dessus.

Après le nom de la fonction viennent ses arguments. Tous les arguments sont obligatoires, sauf ceux explicitement signalés comme optionnels. Les arguments sont ajoutés entre les parenthèses et sont séparés par des points-virgules, sans espace entre les arguments et les points-virgules.

Selon la nature de la fonction, les arguments peuvent être saisis comme ceci :

"donnée texte"	Les guillemets indiquent que du texte ou une chaîne de caractères est saisi.
9	Le nombre 9 est saisi en tant que nombre.
"9"	Le nombre 9 est saisi en tant que texte.
A1	L'adresse de ce qui se trouve dans la cellule A1 est saisie.
'texte'	Une étiquette de colonne ou de ligne, à condition que l'option Outils > Options > LibreOffice Calc > Calcul > Rechercher automatiquement les étiquettes de colonnes et lignes soit cochée.

Fonctions imbriquées

Les fonctions peuvent aussi être utilisées en tant qu'arguments à l'intérieur d'autres fonctions. Elles sont appelées des fonctions imbriquées.

=SOMME(2;PRODUIT(5;7))

Pour avoir une idée de ce que peuvent faire les fonctions imbriquées, imaginez que vous construisez un module d'auto-apprentissage. Durant ce module, les étudiants remplissent trois questionnaires, dont les résultats sont mis dans les cellules A1, A2 et A3. En A4, vous pouvez créer une formule imbriquée qui commence par faire la moyenne des résultats de ce jeu de questions-réponses avec la formule **=MOYENNE(A1:A3)**. La formule utilise alors la fonction SI pour donner aux étudiants un retour correspondant à leur note moyenne. La formule en entier serait :

=SI(MOYENNE(A1:A3)>85;"Bravo ! Vous pouvez avancer au module suivant";"Echec. Veuillez revoir votre cours. Contactez votre enseignant si nécessaire.")

Selon sa moyenne, l'étudiant recevra un message de félicitation ou d'échec.

Notez qu'il ne *faut pas* inclure de signe = devant la fonction imbriquée (MOYENNE en l'occurrence). Le signe qui se trouve au début de la formule est suffisant.

Les exemples fournis sont simples pour expliquer les concepts plus clairement, mais, à travers les fonctions imbriquées, une formule Calc peut vite devenir plus complexe.

Note

Calc vous suggère la syntaxe de la fonction dans une info-bulle lors de la saisie.

Liste de fonctions

Disponible depuis le menu **Insertion**, la Liste des fonctions (Figure 10) va ancrer automatiquement un panneau sur le côté droit de la fenêtre d'édition de Calc. Si vous le souhaitez, vous pouvez appuyer sur **Ctrl+Maj+F10**, ou garder appuyée **Ctrl** et double-cliquer dans une zone vide en haut du panneau, pour supprimer l'ancrage du panneau et le transformer en fenêtre flottante.

La Liste des fonctions comprend une brève description de chaque fonction et de ses arguments ; mettez en surbrillance la fonction et regardez en bas du panneau pour voir cette description. Si nécessaire, placez le curseur au-dessus de la séparation entre la liste et la description ; lorsque le curseur devient une double flèche avec lignes parallèles, faites le glisser pour augmenter la place de la description. Double-cliquez sur le nom d'une fonction pour l'insérer dans la cellule en cours avec des substituants destinés à être remplacés par les valeurs que vous aurez définies pour chacun de ses arguments.




Figure 10 : Liste de fonctions ancrée sur la droite de la fenêtre Calc

Si vous cliquez sur la barre à l'endroit des cinq points et des flèches (indiqué par l'ellipse de la Figure 10), vous masquerez la liste du côté droit de l'écran. Si vous cliquez à nouveau à cet endroit, la liste réapparaîtra. C'est une façon pratique de disposer de la liste pour une utilisation quasiment instantanée.

La manipulation de la Liste des fonctions est presque aussi rapide qu'une entrée manuelle, et a l'avantage de vous affranchir de la mémorisation d'une fonction que vous voulez utiliser. En théorie, cette méthode devrait générer moins d'erreurs. En pratique cependant, certains utilisateurs peuvent commettre des maladresses en renseignant les valeurs dans les substituants. Une autre fonctionnalité est la possibilité d'afficher les dernières fonctions utilisées ou de les afficher par catégorie.

Assistant Fonction

Une méthode de saisie communément utilisée est l'*Assistant Fonction* (Figure 11). Pour ouvrir l'Assistant Fonction, choisissez **Insertion > Fonction**, ou cliquez sur le bouton *fx* de la barre de formule, ou appuyez sur *Ctrl+F2*. Une fois ouvert, l'Assistant Fonction a les mêmes fonctionnalités que la Liste des fonctions, mais comporte en plus des champs dans lesquels vous pouvez voir le résultat de la fonction finale, ainsi que le résultat d'une formule plus vaste dont elle fait partie.

Sélectionnez une catégorie de fonctions pour raccourcir la liste, puis faites défiler pour trouver le nom de la fonction que vous souhaitez. Quand vous sélectionnez une fonction, sa description apparaît du côté droit de la boîte de dialogue. Double-cliquez sur la fonction souhaitée.




Figure 11 : Onglet Fonctions de l'Assistant Fonction après double-clic sur la fonction

L'Assistant affiche alors une zone sur la droite où vous pouvez entrer manuellement des données dans les champs ou cliquer sur le bouton **Sélectionner**  pour réduire l'assistant, afin de pouvoir sélectionner des cellules directement dans la feuille.




Figure 12 : Assistant Fonction après réduction

Pour sélectionner des cellules, cliquez directement dans la cellule ou gardez appuyé le bouton de la souris et faites glisser pour sélectionner la plage voulue.

Une fois la cellule ou la plage sélectionnée, cliquez sur le bouton **Agrandir**  pour retourner dans l'Assistant.

Si plusieurs arguments sont nécessaires, sélectionnez le champ suivant et répétez la procédure de sélection pour la cellule ou plage de cellules suivante. Répétez autant que nécessaire. L'Assistant accepte jusqu'à 30 plages ou arguments pour la fonction SOMME.

Cliquez sur **OK** pour accepter la fonction, l'insérer dans la cellule et obtenir le résultat.

Vous pouvez également sélectionner l'onglet *Structure* (Figure 13) pour voir l'arborescence des parties de la formule. Son principal avantage par rapport à la Liste de fonctions est que chaque

argument se trouve dans son propre champ, ce qui le rend plus facile à gérer. Le coût de cette fiabilité est une saisie plus lente, mais c'est souvent le prix à payer, étant donné que la précision est généralement plus importante que la vitesse pour créer une feuille.




Figure 13 : Onglet Structure de l'Assistant Fonction

Ligne de saisie

Après avoir saisi une formule dans la Ligne de saisie, appuyez sur la touche *Entrée* ou cliquez sur le bouton **Accepter** de la barre de formule pour insérer cette formule dans la cellule et obtenir son résultat.

- 1 Zone de nom listant certaines fonctions courantes (uniquement en cours de saisie d'une formule)
- 2 Assistant Fonctions
- 3 Annuler
- 4 Accepter
- 5 Ligne de saisie




Figure 14 : Barre de formule

Si vous voyez la formule dans la cellule au lieu de voir le résultat, c'est que la cellule est formatée au format *Texte* ou que l'option *Formules* est cochée dans la section *Affichage* de **Outils > Options > LibreOffice Calc > Affichage**. Dans ce cas, décochez *Formules*, et le résultat apparaîtra. Toutefois, vous pouvez toujours voir la formule dans la Ligne de saisie.

Stratégies pour créer un classeur

Si vous manipulez un classeur plus complexe qu'un document à une seule feuille dans Calc, cela vaut la peine de préparer un peu votre travail. Évitez les pièges suivants :

- taper des valeurs fixes dans des formules ;
- l'absence de notes et de commentaires décrivant ce que fait le système, notamment quelles saisies sont requises et d'où viennent les formules (si elles ne sont pas créées de A à Z) ;
- ne pas contrôler les résultats des formules.

Le piège des valeurs fixes

Bien des utilisateurs construisent des formules longues et complexes avec des valeurs fixes saisies directement dans ces formules.

Par exemple, la conversion d'un montant monétaire d'une devise à une autre nécessite la connaissance du taux de change en cours. Si vous saisissez dans la cellule C1 la formule **=0,75*B1** (pour calculer par exemple la valeur en Euros d'un montant en Dollars US situé dans la cellule B1), vous devrez modifier cette formule si le taux de change varie de 0,75 à une autre valeur. Il est plus facile de fixer une cellule de saisie du taux de change et d'y faire référence à chaque fois qu'une formule le requiert. Les calculs de type "et si ?" sont également simplifiés : et si le taux de change varie de 0,75 à 0,70 ou 0,80 ? Il n'y aura alors pas de modifications de formules à effectuer, et le taux utilisé dans les calculs apparaîtra clairement. Le fait d'éclater des formules complexes en plusieurs parties plus facilement gérables contribue également à minimiser les erreurs et à en faciliter la résolution.

Manque de documentation

Le manque de documentation est une erreur communément répandue. Beaucoup d'utilisateurs préparent une feuille simple qui, avec le temps, évolue et devient bien plus complexe. Sans documentation, le propos original et la méthodologie deviennent souvent obscurs et difficiles à déchiffrer. Dans ce cas, il est souvent plus facile de tout recommencer depuis le début, et de gaspiller ainsi le travail effectué précédemment. Si vous insérez des commentaires dans les cellules, si vous utilisez des étiquettes et des en-têtes, un classeur pourra être modifié ultérieurement, par vous ou par d'autres, et beaucoup de temps et d'efforts seront épargnés.

Formules de contrôle d'erreur

Le fait d'ajouter des colonnes de données ou des sélections de cellules dans une feuille provoque souvent des erreurs dues à l'omission de cellules, à la mauvaise spécification de plages de données ou au comptage en double des cellules. Il est utile de mettre en place des contrôles dans vos classeurs. Par exemple, vous avez construit un classeur pour calculer des colonnes de chiffres et avez utilisé SOMME pour calculer le total de chaque colonne. Vous pourrez contrôler ce résultat en ajoutant (dans une colonne non imprimable) un ensemble de totaux de lignes et en les additionnant. Les deux nombres, total des lignes et total des colonnes, devront correspondre. Si ce n'est pas le cas, votre classeur comporte une erreur.

Vous pouvez même construire une formule pour calculer la différence entre les deux totaux et afficher un message d'erreur dans le cas où elle retourne une valeur non nulle (voir Figure 15).

Démonstration du contrôle des erreurs				
Somme des colonnes ColA, ColB et ColC				
	ColA	ColB	ColC	Somme des lignes
	0	0,64	0,02	0,66
	0,43	0,23	0,75	1,41
	0,91	0,57	0,59	2,07
	0,07	0,07	0,45	0,59
	0,37	0,33	0,04	0,74
	0,34	0,06	0,98	1,38
	0,95	0,34	0,65	1,94
	0,93	0,08	0,63	1,64
	0,61	0,82	0,17	1,6
Somme des colonnes	4,61	2,5	=SOMME(E8:E15)	
		TOTAL :	11,37	12,03
				ERREUR !!!

Figure 15 : Contrôle des erreurs des formules

Stratégies pour créer des formules et des fonctions

Les formules qui font plus qu'un simple calcul ou une somme de valeurs de lignes ou de colonnes ont habituellement plusieurs arguments. Par exemple, l'équation classique de la cinématique

$$x = x_0 + v_0 t + \frac{1}{2} a t^2$$

calcule la position d'un corps sachant sa position d'origine, sa vitesse initiale, son accélération et le temps passé pour se déplacer de son état initial à son état final.

Pour faciliter la présentation, c'est une bonne pratique de construire une feuille de manière identique à celle de la Figure 16. Dans cet exemple, les variables individuelles sont saisies dans des cellules de la feuille et aucune modification de formule (dans la cellule B9) n'est nécessaire.

B9 fx Σ = =B3+B4*B6+0,5*B5*B6^2			
	A	B	C
1	Mouvement rectiligne uniformément accéléré		
2			
3	Position d'origine, x0	50 mètres	
4	Vitesse initiale, v0	120 mètres/seconde	
5	Accélération constante, a	2 mètres/seconde ²	
6	Temps, t	12 secondes	
7			
8	Distance entre les positions initiale et finale		
9		x= 1634 mètres	
10			
11	Source :		
12	Equation de cinématique		
13	$x = x_0 + v_0 t + (1/2) a t^2$		

Figure 16 : Construire une formule avec des arguments

Vous pouvez adopter plusieurs approches générales quand vous créez une formule. Au moment de décider de la démarche à adopter, pensez aux autres personnes qui auront à utiliser le

classeur, à l'évolution de ce classeur et aux changements qui pourraient être effectués dans l'utilisation de la formule.

Si d'autres utilisateurs travaillent sur ce classeur, assurez-vous qu'il est facile de comprendre quelle saisie est nécessaire et à quel endroit. L'explication de l'objet du classeur, des bases des calculs, des saisies indispensables et des résultats obtenus est souvent placée dans la première feuille.

Un classeur que vous construisez aujourd'hui, avec nombre de formules compliquées, peut ne pas être aussi évident à utiliser et à maîtriser 6 ou 12 mois plus tard. N'hésitez pas à user de commentaires et de notes afin de documenter votre travail.

Vous-même, en tant que concepteur du classeur, pouvez être conscient que vous ne devez pas utiliser des valeurs négatives ou nulles pour un argument particulier, mais si un autre utilisateur entre de telles valeurs, votre formule sera-t-elle suffisamment robuste, ou retournera-t-elle simplement un message standard (souvent peu utile) ? C'est un bon usage de traquer les erreurs avec des instructions comprenant une forme de logique ou à l'aide du formatage conditionnel.

Placer une formule unique dans chaque cellule

La stratégie la plus basique est de simplifier au maximum ce dont les formules ont besoin. Vous placez alors une formule unique dans chaque cellule appropriée. Cette stratégie ne peut être recommandée que pour les classeurs très simples ou "éphémères" (à usage unique).

Séparer une formule en plusieurs parties, puis les combiner

Cette deuxième stratégie ressemble à la première, mais vous allez séparer les longues formules en plusieurs parties plus petites, puis combiner ces parties en un tout. Beaucoup d'exemples de ce type existent pour les calculs scientifiques et industriels, où les résultats intermédiaires sont stockés dans plusieurs endroits du classeur.

Dans tous les cas, vous devriez adopter les principes de base de création de formules décrits précédemment.

Utiliser l'éditeur Basic pour créer des fonctions

Une troisième stratégie est d'utiliser l'éditeur Basic et de créer vos propres fonctions et macros. Cette approche peut être utilisée pour simplifier l'utilisation du classeur par l'utilisateur final et garder des formules simples avec un risque d'erreur limité. Cette approche peut également rendre la maintenance plus facile en centralisant les modifications et mises à jour en un seul endroit. L'utilisation des macros décrite dans le Chapitre 12 est un sujet à part entière. Le danger de la sur-utilisation des macros et des fonctions personnalisées réside dans le fait que les principes sur lesquels le classeur se base peuvent devenir plus difficiles à appréhender par un utilisateur autre que l'auteur d'origine (et parfois même par l'auteur !).

Détecter et corriger les erreurs

Il est ordinaire de se trouver dans des situations où des erreurs sont affichées. Même avec tous les outils disponibles dans Calc pour vous aider à entrer des formules, commettre des erreurs est fréquent. Beaucoup de personnes trouvent la saisie de nombres difficile et peuvent perpétrer des erreurs au sujet du type d'entrée qu'un argument d'une fonction nécessite. En plus de corriger les erreurs, vous pouvez vouloir trouver les cellules utilisées dans une formule pour changer leurs valeurs ou vérifier la réponse.

Calc dispose de trois outils pour réaliser des investigations dans les formules et les cellules qu'elles référencent : les messages d'erreur, le codage en couleur et l'Audit.

Messages d'erreur

Le plus simple pour gérer les erreurs est d'utiliser les messages d'erreur. Les messages d'erreurs s'affichent à la place du résultat dans la cellule d'une formule ou dans l'Assistant Fonction.

Un message d'erreur pour une formule est habituellement un nombre à trois chiffres de 501 à 532, ou parfois un bout de texte comme #NOM?, #REF ! ou #VALEUR !. Le numéro de l'erreur apparaît généralement dans la cellule et une courte explication de l'erreur est affichée dans la barre d'état.

La plupart des messages d'erreur indiquent un problème avec la façon dont la formule est saisie, mais d'autres peuvent indiquer que vous avez dépassé une limite de Calc ou de son paramétrage en cours.

Les messages d'erreurs ne sont pas toujours conviviaux et peuvent intimider les nouveaux utilisateurs. Toutefois, ce sont des indices utiles pour corriger les erreurs. Vous pouvez trouver des explications détaillées pour chacun d'entre eux dans l'Aide, en recherchant dans l'index *Message d'erreur ###*, ou dans l'Annexe C. Quelques messages parmi les plus fréquents sont expliqués dans la table suivante.

#####	Il ne s'agit pas d'une erreur, mais simplement d'une colonne qui n'est pas assez large pour afficher le résultat.
#NOM? (525)	L'identificateur (de nom, d'étiquette de ligne ou de colonne, de macro...) est inconnu.
#REF ! (524)	La ligne, colonne ou feuille référencée n'existe pas.
#VALEUR ! (519)	La valeur de l'un des arguments n'est pas du type correct. Par exemple, des guillemets manquent autour d'une valeur, ou une cellule référencée contient un texte au lieu d'un nombre.
509	Un opérateur manque dans la formule.
510	Un argument manque dans la formule.
#DIV/0 ! (532)	Division par zéro.

Exemples d'erreurs fréquentes

#DIV/0 ! (532) Division par zéro

L'erreur résulte de la division d'un nombre par le nombre zéro (0) ou par une cellule vide. Vous pouvez éviter facilement ce type de problème. Utilisez une fonction conditionnelle testant une valeur zéro ou une cellule vide. La Figure 17 montre la division de la colonne B par la colonne C, avec l'apparition de deux erreurs dues à un zéro ou une cellule vide dans la colonne C.

	A	B	C	D
1				
2	Régions	Population	Magasins	Population moyenne / Magasin
3	Alsace	1 851 443	12	154 287
4	Aquitaine	3 231 860	27	119 699
5	Auvergne	1 345 712	0	#DIV/0 !
6	Basse-Normandie	1 473 946	7	210 564
7	Bourgogne	1 644 756	3	548 252
8	Bretagne	3 197 975	15	213 198
9	Centre	2 544 791	23	110 643
10	Champagne-Ardenne	1 336 242	2	668 121
11	Corse	309 339		#DIV/0 !
12	Franche-Comté	1 172 549	7	167 507

Figure 17 : Exemples d'erreurs #DIV/0 ! Division par zéro

Il est très fréquent de rencontrer de telles erreurs dans une situation où des données sont manquantes ou ne sont pas saisies correctement. Quand une telle possibilité existe, une fonction SI peut être utilisée pour afficher les données correctement. La formule **=SI(C3>0;B3/C3;"Pas de valeur")** peut être saisie. Cette formule peut alors être copiée dans le reste de la colonne D. La signification littérale de cette formule serait : Si C3 est supérieur à 0, alors divise B3 par C3, sinon affiche "Pas de valeur".

	A	B	C	D	E
1					
2	Régions	Population	Magasins	Population moyenne / Magasin	
3	Alsace	1 851 443	12	=SI(C3>0;B3/C3;"Pas de valeur")	
4	Aquitaine	3 231 860	27	119 699	
5	Auvergne	1 345 712	0	Pas de valeur	
6	Basse-Normandie	1 473 946	7	210 564	
7	Bourgogne	1 644 756	3	548 252	
8	Bretagne	3 197 975	15	213 198	
9	Centre	2 544 791	23	110 643	
10	Champagne-Ardenne	1 336 242	2	668 121	
11	Corse	309 339		Pas de valeur	
12	Franche-Comté	1 172 549	7	167 507	

Figure 18 : Solution pour la division par zéro

Il est également possible, dans le dernier argument, d'utiliser un nombre standard à utiliser par défaut (0 par exemple). Une pratique courante est d'utiliser dans ce cas, des doubles guillemets ("") pour afficher un texte vide. Attention à ne recourir à cette solution que pour des cellules ne devant pas être considérées comme des nombres dans des formules, sous peine de déclencher une erreur #VALEUR ! (voir ci-dessous).

#VALEUR ! Valeur incorrecte et #REF ! Référence incorrecte

L'erreur #VALEUR ! est également très fréquente, par exemple quand une cellule contient une donnée d'un type incorrect. Dans l'exemple de la Figure 19, le texte "Inconnu" a été saisi dans la cellule C8, alors que la formule de la colonne D attend à cet endroit un nombre.

	A	B	C	D
1				
2	Régions	Population	Magasins	Population moyenne / Magasin
3	Alsace	1 851 443	12	154 287
4	Aquitaine	3 231 860	27	119 699
5	Auvergne	1 345 712	0	Pas de valeur
6	Basse-Normandie	1 473 946	7	210 564
7	Bourgogne	1 644 756	3	548 252
8	Bretagne	3 197 975	Inconnu	#VALEUR !
9	Centre	2 544 791	23	110 643
10	Champagne-Ardenne	1 336 242	2	668 121
11	Corse	309 339		Pas de valeur
12	Franche-Comté	1 172 549	7	167 507

Figure 19 : Saisie incorrecte provoquant l'erreur #VALEUR !

L'erreur #REF ! est provoquée par une référence manquante. Dans l'exemple de la Figure 20, la formule fait référence à une feuille qui a été supprimée.

SOMME			  	=SOMME(Feuille2.A1:A4)	
	A	B	C	D	
1					
2	=SOMME(Feuille2.A1:A4)				
3					

A2			  =	=SOMME(#REF!.A1:A4)	
	A	B	C	D	
1					
2	#REF !				
3					

Figure 20 : Feuille supprimée provoquant l'erreur #REF !

Note

Pour éviter les erreurs #VALEUR ! et #REF !, il est souvent utile de donner à une cellule, comme B5 dans la Figure 22, un nom comme Total2010. Ainsi, le programme va transporter ce nom durant la copie de la formule, et cela évite d'employer une référence absolue (\$B\$5).

Codage en couleur

Un autre outil utile pour la révision des formules est le codage en couleur. Quand vous éditez une formule déjà saisie, les cellules ou les plages utilisées dans chaque argument ont un contour en couleur.

Calc utilise huit couleurs pour entourer les cellules référencées, en commençant par le bleu pour la première cellule, puis en continuant par le rouge, le magenta, le vert, le bleu marine, le marron, le violet et le jaune, et en répétant cette séquence autant de fois que nécessaire.

SOMME fx ✖ ✓ =SI(C3>0;B3/C3;"Pas de valeur")					
	A	B	C	D	E
1					
2	Régions	Population	Magasins	Population moyenne / Magasin	
3	Alsace	1 851 443	12	=SI(C3>0;B3/C3;"Pas de valeur")	

Figure 21 : Codage en couleur

L'Audit

Dans les classeurs longs ou complexes, le codage en couleur devient moins utile. Dans ce cas, vous pouvez utiliser le sous-menu dans **Outils > Audit**. L'Audit est un outil pour vérifier quelles sont les cellules utilisées dans les arguments d'une formule (antécédents), ou dans quelles autres formules elle est utilisée (dépendants). Il peut également être utilisé pour repérer les erreurs, marquer les données incorrectes (les informations dans les cellules qui n'ont pas le format correct pour l'argument d'une fonction), ou supprimer les repères des antécédents et des dépendants.

Pour utiliser l'Audit, sélectionnez une cellule contenant une formule, puis démarrez l'Audit. Dans le classeur, vous verrez des lignes se terminant par des cercles pour indiquer les antécédents et des lignes se terminant par des flèches pour les dépendants. Ces lignes montrent le flux des informations. Double-cliquez sur le repère pour sélectionner la cellule référencée.

Utilisez l'Audit pour vous aider à suivre les antécédents référencés par une formule dans une cellule. En suivant ces antécédents, vous pourrez souvent trouver la source des erreurs. Placez le curseur dans la cellule en question, puis choisissez **Outils > Audit > Repérer les antécédents** dans la barre de menus ou appuyez sur **Maj+F7**. Les Figures 22, 23 et 24 montrent un exemple simple de repérage des antécédents.

C5 fx Σ = =SOMME(C6:C15)					
	A	B	C	D	E
1	Estimation de population au 1 ^{er} janvier, par région, sexe et grande classe d'âge				
2					
3		Population			
4	Régions	2010	2005	Evolution	
5	France – Nord	33 363 138	28 785 076	15,9%	
6	Alsace	1 851 443	1 803 402	2,7%	
7	Basse-Normandie	1 473 946	1 452 425	1,5%	
8	Bretagne	3 197 975	3 066 585	4,3%	
9	Champagne-Ardenne	1 336 242	1 339 400	-0,2%	
10	Haute-Normandie	1 837 388	1 806 677	1,7%	
11	Île-de-France	11 797 021	11 442 143	3,1%	
12	Lorraine	2 352 324	2 332 468	0,9%	
13	Nord - Pas-de-Calais	4 035 269	4 015 613	0,5%	
14	Pays de la Loire	3 566 686	3 415 391	4,4%	
15	Picardie	1 914 844	-1 889 028	-201,4%	
16					
17	Source : Insee - Estimations de population (résultats provisoires arrêtés fin 2011).				

Figure 22 : Audit > Repérer les antécédents - Se placer dans la cellule




Figure 23 : Audit > Repérer les antécédents - Choisir l'option

C5					=SOMME(C6:C15)
	A	B	C	D	E
1	Estimation de population au 1 ^{er} janvier, par région, sexe et grande classe d'âge				
2					
3		Population			
4	Régions	2010	2005	Evolution	
5	France – Nord	33 363 138	28 785 076	15,9%	
6	Alsace	1 851 443	1 803 402	2,7%	
7	Basse-Normandie	1 473 946	1 452 425	1,5%	
8	Bretagne	3 197 975	3 066 585	4,3%	
9	Champagne-Ardenne	1 336 242	1 339 400	-0,2%	
10	Haute-Normandie	1 837 388	1 806 677	1,7%	
11	Île-de-France	11 797 021	11 442 143	3,1%	
12	Lorraine	2 352 324	2 332 468	0,9%	
13	Nord - Pas-de-Calais	4 035 269	4 015 613	0,5%	
14	Pays de la Loire	3 566 686	3 415 391	4,4%	
15	Picardie	1 914 844	-1 889 028	-201,4%	
16					
17	Source : Insee - Estimations de population (résultats provisoires arrêtés fin 2011).				

Figure 24 : Audit > Repérer les antécédents - Résultat

Vous vous interrogez sur le nombre affiché dans la cellule C5, qui vous paraît erroné. La cause peut être détectée dans les cellules entourées. Dans ce cas, la cellule C15 contient par erreur un signe négatif. Supprimer ce signe résout tout simplement le problème.

Dans d'autres cas, vous devrez tracer l'erreur. Utilisez la fonction **Outils > Audit > Repérer les erreurs** pour repérer les cellules à l'origine de l'erreur.

Exemples de fonctions

Pour les novices, les fonctions sont l'une des fonctionnalités les plus intimidantes de LibreOffice Calc. Les nouveaux utilisateurs apprennent rapidement que les fonctions sont des ressources importantes des classeurs. On peut en dénombrer plus de 350, et beaucoup d'entre elles nécessitent des paramètres qui reposent sur un savoir spécialisé. Heureusement, Calc comprend un certain nombre de fonctions à la portée de tous.

Fonctions de base arithmétiques et statistiques

Les fonctions les plus simples créent des formules d'arithmétique de base ou pour calculer des nombres dans une plage de cellules.

Arithmétique de base

Les fonctions de base de l'arithmétique sont l'addition, la soustraction, la multiplication et la division.

Certaines de ces opérations disposent de leur propre fonction :

- **SOMME** pour l'addition ;
- **PRODUIT** pour la multiplication.

Cependant, pour les opérations simples, beaucoup d'utilisateurs préfèrent les symboles traditionnels, en utilisant le signe plus (+) pour l'addition, le signe moins (-) pour la soustraction, l'astérisque (*) pour la multiplication et la barre oblique (/) pour la division. Ces symboles sont plus rapides à saisir.

Un choix identique est disponible si vous voulez élever un nombre à la puissance d'un autre. Au lieu de saisir **=PUISSANCE(A1; 2)**, vous pouvez saisir **=A1^2**.

De plus, les symboles ont l'avantage de vous faire saisir les formules dans l'ordre habituel du raisonnement humain, plutôt que l'ordre du traitement informatique employé par sa fonction équivalente. Par exemple, au lieu de saisir **=SOMME(A1:A2)**, ou **=SOMME(A1; A2)**, vous saisissez **=A1+A2** (voir l'astuce suivante). Cet ordre est souvent préférable pour les opérations composées, quand saisir **=A1*(A2+A3)** est plus court et plus facile à lire que **=PRODUIT(A1; SOMME(A2:A3))**.

Le principal inconvénient de l'utilisation des opérateurs arithmétiques réside dans le fait que vous ne pouvez pas utiliser directement une plage de cellules. En d'autres termes, pour saisir l'équivalent de **=SOMME(A1:A4)**, vous devrez saisir **=A1+A2+A3+A4**.

Un autre inconvénient est que si l'on insère par exemple une cellule ou une ligne entre A1 et A4, une formule utilisant les opérateurs n'inclura pas la nouvelle cellule contrairement à une formule utilisant une plage.

Astuce

La fonction SOMME étant très utilisée, la barre de formule comprend un bouton insérant cette fonction dans la cellule courante. LibreOffice suggère automatiquement une plage de cellules, au-dessus ou à gauche. Si la plage contient des filtres, la fonction SOUS.TOTAL est insérée à la place de la fonction SOMME.

Statistiques de base

Une autre utilisation fréquente des fonctions dans un classeur est de tirer des informations d'une liste, comme la série de notes d'une classe, ou les ventes par produit d'une société pour un trimestre.

Vous pouvez bien sûr parcourir une liste de nombres si vous voulez des informations simples, comme le nombre le plus haut, le plus bas ou la moyenne, par exemple. Le souci est que, plus la liste s'allonge, plus vous allez y passer du temps et plus le risque d'erreur augmente. Il est donc plus rapide et plus efficace de saisir une fonction. Ceci explique l'existence d'une fonction comme **NB**, qui ne fait que compter combien de nombres sont présents dans la plage de cellules spécifiée.

De même, pour trouver le nombre le plus haut ou le plus bas, vous pouvez utiliser **MIN** ou **MAX**. Pour chacune de ces fonctions, tous les arguments peuvent être soit une plage de cellules, soit une série de cellules entrées individuellement.

Chacune a une fonction proche, **MINA** ou **MAXA**, qui fait la même chose, mais traite les cellules formatées en texte comme ayant une valeur 0 (d'autres fonctions avec un "A" supplémentaire à la fin existent). Ce peut être utile si vous utilisez une note de texte pour indiquer, par exemple, qu'un élève était absent à un examen et que vous voulez vérifier si un examen complémentaire est nécessaire.

Pour plus de souplesse pour des opérations similaires, vous pouvez utiliser les fonctions **GRANDE.VALEUR** ou **PETITE.VALEUR**, qui comportent toutes les deux un argument supplémentaire pour le rang. Si un rang de 1 est utilisé avec **GRANDE.VALEUR**, vous obtiendrez le même résultat qu'avec **MAX**. Si le rang est 2, le résultat sera alors le deuxième résultat le plus grand. De même, un rang de 2 utilisé avec **PETITE.VALEUR** vous donnera le deuxième nombre le plus petit. **GRANDE.VALEUR** et **PETITE.VALEUR** seront utiles pour un contrôle permanent et, en changeant l'argument rang, vous pourrez rapidement obtenir plusieurs résultats.

Vous devriez être un expert si vous voulez trouver la loi de Poisson d'un échantillon, ou les lois normale ou normale inverse (et, si vous l'êtes, vous trouverez des fonctions dans Calc pour ce genre de choses). Cependant, pour les autres, il y a des fonctions statistiques plus simples que vous pouvez rapidement apprendre à utiliser.

En particulier, si vous avez besoin d'une moyenne, vous avez le choix entre plusieurs fonctions. Vous pouvez trouver la moyenne arithmétique, c'est-à-dire la somme des valeurs d'une liste divisée par le nombre de ces valeurs, en utilisant **MOYENNE**, ou **MOYENNEA** pour inclure les valeurs texte et leur donner une valeur de zéro.

Vous pouvez de plus obtenir d'autres informations de l'ensemble de données :

- **MEDIANE** : la valeur de la liste qui se trouve exactement à mi-chemin entre le nombre le plus grand et le nombre le plus petit.
- **MODE** : la valeur la plus fréquente dans la liste.
- **QUARTILE** : la valeur à une position déterminée dans un tableau de nombres. Après la plage de cellules, vous devez saisir le type de quartile : 0 pour la valeur minimale (équivalent à **MIN**), 1 pour la valeur à 25 %, 2 pour la valeur à 50 %, 3 pour la valeur à 75 % et 4 pour la valeur maximale (équivalent à **MAX**). Notez que le résultat pour les types de 1 à 3 peut ne pas correspondre à une valeur de la liste.
- **RANG** : la position d'une valeur de la liste donnée dans la liste entière, en suivant les valeurs croissantes ou décroissantes. Vous devez saisir l'adresse de la cellule dont vous cherchez le rang, la plage de cellules et le type de rang (ordre décroissant si le type est 0 ou vide, ordre croissant sinon).

Certaines de ces fonctions se recouvrent ; par exemple, **MIN** et **MAX** peuvent également être obtenues avec **QUARTILE**. Dans d'autres cas, un tri ou un filtre personnalisés peuvent donner le même résultat. Votre choix dépendra de votre tempérament et de vos besoins. Certains

préféreront utiliser MIN et MAX parce que c'est plus facile à retenir, tandis que d'autres préféreront QUARTILE parce que c'est plus puissant.

Utiliser ces fonctions

Dans certains cas, vous pouvez obtenir des résultats similaires à certaines de ces fonctions en paramétrant un filtre ou un tri personnalisés. Cependant, en général, les fonctions se révèlent être plus facilement ajustables que les filtres ou les tris et comportent une plus grande variété de possibilités.

Certaines fois, vous pouvez vouloir entrer temporairement une ou plusieurs formules dans une cellule vide qui convient, et les supprimer lorsque vous avez terminé. Cependant, si vous vous apercevez que vous utilisez toujours les mêmes formules, vous pouvez songer à créer un modèle en laissant de l'espace pour toutes les formules que vous utilisez et en utilisant la cellule à leur gauche pour saisir une étiquette. Une fois le modèle créé, vous pouvez facilement mettre à jour chaque formule après avoir saisi des valeurs, soit automatiquement, soit "à la volée" en appuyant sur la touche *F9* pour actualiser toutes les cellules sélectionnées.

Arrondir les nombres

Dans un but statistique ou mathématique, Calc dispose de plusieurs façons d'arrondir les nombres. Si vous êtes programmeur, vous serez familier avec certaines de ces méthodes. Toutefois, vous n'avez pas besoin d'être un spécialiste pour trouver ces méthodes utiles. Vous pouvez vouloir arrondir pour des raisons monétaires, ou parce que les chiffres décimaux ne sont pas judicieux. Par exemple, si les articles que vous commandez sont conditionnés par 100, le fait que vous n'en ayez besoin que de 66 au maximum est hors sujet pour vous ; vous devrez arrondir ce nombre pour pouvoir commander. En connaissant les options d'arrondi, vous pourrez rendre vos classeurs immédiatement plus opérationnels.

Attention



Vous pouvez utiliser l'option **Outils > Options > LibreOffice Calc > Calcul > Limiter les décimales pour le format de nombre général** ou le champ *Décimales* de l'option **Format > Cellules > Nombres** pour arrondir un nombre. Cependant, soyez bien conscient que l'arrondi ne s'applique qu'à l'affichage de la cellule, et que son contenu reste inchangé. Par exemple, si la valeur de la cellule A1 est de 1,2345 et que vous la formatez avec deux décimales, 1,23 s'affichera dans A1. Toutefois, si, dans la cellule A2 avec le format Standard, vous saisissez la formule **=A1**, c'est bien 1,2345 qui s'affichera.

Si vous utilisez une fonction d'arrondi, vous avez deux possibilités pour construire vos formules. Vous pouvez imbriquer un calcul dans une fonction d'arrondi. Par exemple, la formule **=ARRONDI (SOMME (A1 : A2))** additionne les nombres des cellules A1 et A2, puis arrondit le résultat à l'entier le plus proche. Cependant, même si vous n'avez pas besoin du résultat exact tous les jours, vous pouvez vouloir vous y référer occasionnellement. Si c'est le cas, vous feriez alors mieux de séparer les deux fonctions, en plaçant **=SOMME (A1 : A2)** dans la cellule A3, **=ARRONDI (A3)** dans A4, et en mettant une étiquette explicite pour chacune des cellules.

Méthodes d'arrondi

La fonction de base pour arrondir les nombres dans Calc est **ARRONDI**. Cette fonction va arrondir un nombre selon les règles habituelles de l'arrondi arithmétique symétrique : si on considère des nombres avec une seule position décimale, une position décimale de 0,4 ou moins va être arrondie au nombre inférieur, tandis qu'une position décimale de 0,5 ou plus va être arrondie au nombre supérieur. Cependant, vous pouvez parfois ne pas vouloir suivre ces règles. Par exemple, si vous êtes un fournisseur qui facture une heure pleine pour toute fraction d'heure travaillée, vous voudrez toujours arrondir au nombre supérieur pour ne pas perdre d'argent. Inversement, vous pouvez décider d'arrondir au nombre inférieur pour consentir une petite remise à un client régulier.

Dans ces cas, vous pouvez préférer utiliser **ARRONDI.SUP** ou **ARRONDI.INF** qui, comme leur nom l'indique, arrondissent un nombre à son entier supérieur ou inférieur le plus proche.

Ces trois fonctions ne nécessitent qu'un argument numérique, la cellule ou le nombre à arrondir. Utilisées uniquement avec cet argument, les trois fonctions effectuent un arrondi à l'entier le plus proche. Ainsi, 46,5 sera arrondi à 47 avec **ARRONDI** ou **ARRONDI.SUP**, et à 46 avec **ARRONDI.INF**. Cependant, si vous utilisez l'argument facultatif, vous pouvez spécifier le nombre de positions décimales à prendre en compte. Si l'argument est de 1 position décimale, alors 48,65 sera arrondi à 48,7 avec **ARRONDI** ou **ARRONDI.SUP**, et à 48,6 avec **ARRONDI.INF**.

Comme alternative à **ARRONDI.INF** quand vous travaillez avec des décimales, vous pouvez utiliser **TRONQUE**. Cette fonction a exactement les mêmes arguments que **ARRONDI.INF**, et vous pouvez donc utiliser l'une ou l'autre fonction indifféremment. Si vous ne travaillez pas avec des décimales, vous pouvez utiliser **ENT** (pour entier) qui n'accepte qu'un argument.

Il est également possible d'utiliser la paire de fonctions **PAIR** et **IMPAIR**. **PAIR** arrondit un nombre positif au nombre entier pair immédiatement supérieur, et un nombre négatif au nombre entier pair immédiatement inférieur. **IMPAIR** fait la même chose avec un nombre entier impair.

Il existe également les fonctions **PLAFOND** et **PLANCHER**. Comme leur nom le laisse deviner, **PLAFOND** arrondit vers le haut et **PLANCHER** vers le bas. Pour ces deux fonctions, le nombre auquel est arrondie la valeur du premier argument est le multiple le plus proche du deuxième argument appelé précision. Par exemple, si votre assurance professionnelle est calculée à la semaine travaillée, le fait que vous n'ayez été ouvert que trois jours durant une semaine n'est pas significatif pour vos dépenses ; vous serez facturé pour une semaine entière, et donc vous pouvez utiliser **PLAFOND** pour calculer vos dépenses du mois. Si la cellule E1 contient la valeur 46,7, **=PLAFOND(E1;7)** va retourner la valeur 49.

Après le nombre et la précision, **PLAFOND** et **PLANCHER** comportent un argument facultatif appelé mode. Si le mode est à 0, et que le nombre et la précision sont des nombres négatifs, le résultat est arrondi vers le haut ; si le mode a une autre valeur, et que le nombre et la précision sont des nombres négatifs, le résultat est arrondi vers le bas. Autrement dit, la fonction **=PLAFOND(-11;-5;0)** va retourner la valeur -10 et la fonction **=PLAFOND(-11;-5;1)** va retourner la valeur -15.

Toutefois, si vous échangez des classeurs entre Calc et MS Excel, ayez en mémoire que l'argument mode n'est pas supporté par Excel. Si vous voulez que le comportement soit cohérent entre les deux tableurs, mettez une valeur non nulle dans le mode.

La fonction **ARRONDI.AU.MULTIPLE** est presque similaire à **PLAFOND** et **PLANCHER**. Comme ces deux fonctions, **ARRONDI.AU.MULTIPLE** nécessite deux arguments, bien que le second soit intitulé multiple au lieu de précision même s'ils sont identiques, ce qui peut apporter une certaine confusion. La différence entre **ARRONDI.AU.MULTIPLE** et **PLAFOND** et **PLANCHER** est que **ARRONDI.AU.MULTIPLE** utilise un arrondi arithmétique symétrique. Ainsi, **=ARRONDI.AU.MULTIPLE(77;5)** retournera la valeur 75 et **=ARRONDI.AU.MULTIPLE(78;5)** retournera la valeur 80.

Attention



Si vous travaillez avec plusieurs positions décimales, ne soyez pas surpris si vous ne voyez pas le même nombre de décimales dans le classeur que ce que vous aviez paramétré dans une fonction d'arrondi. Une limite peut exister dans le champ *Limiter les décimales pour le format de nombre général* dans **Outils > Options > LibreOffice Calc > Calcul**. Paramétrez alors cette limite comme vous le souhaitez, ou décochez cette option.

Ce problème ne se pose pas si vous utilisez un style de cellule dans lequel vous pouvez définir entre autres le nombre de décimales (voir le chapitre 4 Utiliser les styles et les modèles dans Calc).


Utiliser les caractères génériques dans les fonctions

Certaines fonctions de Calc autorisent l'utilisation de caractères génériques : SOMME.SI, NB.SI, EQUIV, CHERCHE, RECHERCHE, RECHERCHEH, RECHERCHEV, BDNB, BDNBVAL, BDSOMME, BDPRODUIT, BDMAX, BDMIN, BDMOYENNE, BDECARTYPE, BDECARTYPEP, BDVAR, BDVARP, BDLIRE.

Les caractères génériques peuvent être employés si la case *Autoriser les caractères génériques dans les formules* est cochée, dans l'option **Outils > Options > LibreOffice Calc > Calcul**. Vous pouvez trouver une explication concernant les caractères génériques dans la section Utiliser les caractères génériques (expressions régulières) du Chapitre 2 (Entrer, modifier et formater des données).

Par exemple, si *Autoriser les caractères génériques dans les formules* est cochée, **=NB.SI(A1:A6;"r.d")** va prendre en compte les cellules dans A1:A6 qui contiennent rad et ROD.

De plus, si la case *Critères de recherche = et <> doivent correspondre à des cellules entières* n'est pas cochée, les cellules *Fred*, *bride* et *Rideau* seront également prises en compte. Si ce paramètre est coché, il peut être contourné en modifiant l'expression : **=NB.SI(A1:A6;"*.r.d.*")**.



	A	B	C	D
1	Fred			
2	rad			
3	ROD			
4	bride			
5	kilo			
6	Rideau			
7		5		
8				

Figure 25 : Utiliser la fonction NB.SI

Les recherches par caractères génériques à l'intérieur des fonctions sont toujours insensibles à la casse, que la case *Respecter la casse* de l'option **Outils > Options > LibreOffice Calc > Calcul** soit cochée ou non. *rad* et *ROD* seront toujours pris en compte comme dans l'exemple de la Figure 25. Cette insensibilité à la casse s'applique également aux structures d'expressions régulières ([[:lower:]] et ([[:upper:]]).

Les caractères génériques ne fonctionnent pas comme de simples comparaisons. Par exemple, **A1="r.d"** retournera FAUX si A1 contient *rad*, même si les caractères génériques sont autorisés. Il ne retournera VRAI que si A1 contient *r.d* (r suivi d'un point suivi de d). Si vous voulez tester des expressions avec des caractères génériques, essayez la fonction NB.SI : **NB.SI(A1 ; "r.d")** retournera 1 ou 0, interprété comme VRAI ou FAUX dans des formules telles que **=SI(NB.SI(A1 ; "r.d"); "Oui"; "Non")**.

Si vous activez l'option *Autoriser les caractères génériques dans les formules*, cela signifie que pour toutes les fonctions ci-dessus, si vous avez besoin des caractères génériques spéciaux (comme les parenthèses) en tant que caractères ordinaires d'une chaîne dans une formule, vous devrez les faire précéder par une barre oblique inversée (\). Ces barres obliques inversées devront être supprimées si, par la suite, l'option *Autoriser les caractères génériques dans les formules* est désactivée.

Fonctions avancées

Comme c'est le cas avec d'autres tableurs, LibreOffice Calc peut être amélioré avec des fonctions définies par les utilisateurs ou des modules externes. La construction de fonctions personnalisées peut être effectuée avec l'éditeur Basic EDI ou en écrivant des modules externes ou des extensions séparées.

Les bases de l'écriture et de l'exécution des macros sont traitées Chapitre 12 (Macros Calc). Les macros peuvent être liées à des menus ou à des barres d'outils pour faciliter leur utilisation, ou être stockées dans des modules standard afin que les fonctions puissent être disponibles dans d'autres documents.

Les modules externes de Calc sont des extensions bureautiques spécialisées qui peuvent étendre les fonctionnalités de LibreOffice avec de nouvelles fonctions Calc intégrées. L'écriture des modules externes nécessite la connaissance du langage C++ et du SDK de LibreOffice, et est réservée aux programmeurs expérimentés. Un certain nombre d'extensions pour Calc ont été écrites et peuvent être trouvées sur le site des [Extensions de LibreOffice](#). Reportez-vous au Chapitre 14 (Paramétrer et personnaliser Calc) pour plus de détails.